Εισαγωγή στη Maple

Εδώ δίνουμε κάποιες χρήσιμες εντολές του προγράμματος Maple.
Η εντολή ***expand*** μας βοηθά να αναπτύξουμε μια αλγεβρική ή τριγωνομετρική παράσταση.

Παραδείγματα:

```
> expand((x+1)^8);
```
$$x^8+8\,x^7+28\,x^6+56\,x^5+70\,x^4+56\,x^3+28\,x^2+8\,x+1$$

```
> expand((x*y+y^2)^2+x);
```
$$x^2\,y^2+2\,x\,y^3+y^4+x$$

```
> sin(2*x):  %=expand(%);
```
$$\sin(2\,x)=2\sin(x)\cos(x)$$

```
> cos(2*x):  %=expand(%);
```
$$\cos(2\,x)=2\cos(x)^2-1$$

```
> tan(2*x):  %=expand(%);
```
$$\tan(2\,x)=\frac{2\tan(x)}{1-\tan(x)^2}$$

```
> sin(x+y):  %=expand(%);
```
$$\sin(x+y)=\sin(x)\cos(y)+\cos(x)\sin(y)$$

```
> cos(x-y):  %=expand(%);
```
$$\cos(x-y)=\cos(x)\cos(y)+\sin(x)\sin(y)$$

```
> tan(x+y):  %=expand(%);
```
$$\tan(x+y)=\frac{\tan(x)+\tan(y)}{1-\tan(x)\tan(y)}$$

```
> ln(x*y):  %=expand(%) assuming positive;
```
$$\ln(x\,y)=\ln(x)+\ln(y)$$

```
> x^(y+z):  %=expand(%);
```
$$x^{(y+z)}=x^y\,x^z$$

```
> (-x)^y:  %=expand(%) assuming x>0;
```
$$(-x)^y=x^y\,(-1)^y$$

Στο επόμενο παράδειγμα, θα αναπτύξουμε το πολυώνυμο $(x+1)^4$ στο χώρο Z_4 ο οποίος αποτελείται από τέσσερα στοιχεία, το 0 , 1, 2 και το 3.

```
> expand((x+1)^4) mod 4;
```

$$x^4 + 2\,x^2 + 1$$

Η εντολή ***combine*** κάνει το αντίθετο από ότι η προηγούμενη εντολή.

Παραδείγματα:

```
> 2*sin(x)*cos(x):  %=combine(%);
```

$$2\sin(x)\cos(x) = \sin(2\,x)$$

```
> 2*sin(x)*cos(y):  %=combine(%);
```

$$2\sin(x)\cos(y) = \sin(x+y) + \sin(x-y)$$

```
> sin(x)^3:  %=combine(%);
```

$$\sin(x)^3 = -\frac{1}{4}\sin(3\,x) + \frac{3}{4}\sin(x)$$

Στο επόμενο παράδειγμα θεωρούμε την συνάρτηση $f = e^{cos^2x+sin^2}$. Η εντολή combine χρησιμοποιεί την τριγωνομετρική ταυτότητα $cos^2x + sin^2x = 1$ και απλοποιεί τη συνάρτηση σε e.

```
> f := exp(cos(x)^2)*exp(sin(x)^2):
> f=combine(f);
```

$$e^{(\cos(x)^2)}\, e^{(\sin(x)^2)} = e$$

Χρησιμοποιώντας το trig στην εντολή, τότε τα cos^2x και sin^2x αντικαθιστώνται με τριγωμετρικές ταυτότητες που περιέχουν διπλάσια γωνιά.

```
> f=combine(f, trig);
```

$$e^{(\cos(x)^2)}\, e^{(\sin(x)^2)} = e^{(1/2\cos(2\,x)+1/2)}\, e^{(1/2-1/2\cos(2\,x))}$$

```
> 3*ln(2)+3*ln(x)-4*ln(5):  %=combine(%);
```

$$3\ln(2) + 3\ln(x) - 4\ln(5) = 3\ln(x) + \ln(\frac{8}{625})$$

Στο επόμενο παράδειγμα θεωρούμε την συνάρτηση $f = e^{x+nlny}$.

```
> f := exp(x+n*ln(y));
```

$$f := e^{(x+n\ln(y))}$$

Αν θέλουμε η Maple να κάνει πράξεις/ιδιότητες λογαρίθμων, όπως $vlnA \rightarrow lnA^v$, βάζουμε στην εντολή το symbolic. Η εντολή anything εφαρμόζει τις ιδιότητες των λογαρίθμων σε οποιαδήποτε παράμετρο ή αριθμό εμφανίζεται στην συνάρτηση.

```
> f =combine(f, ln, anything, symbolic);
```

$$e^{(x+n\ln(y))} = e^{(x+\ln(y^n))}$$

```
> combine(%);
```

$$e^{(x+n\ln(y))} = y^n\, e^x$$

```
> 2^(1/3)*(x+1)^(1/3):  %=combine(%);
```

$$2^{(1/3)}\,(x+1)^{(1/3)} = (2\,x+2)^{(1/3)}$$

Μερικές φορές θέλουμε να απλοιήσουμε μια παράσταση. Σε αυτή την περίπτωση χρησιμοποιούμε την εντολή ***simplify***.

Παραδείγματα:

```
> simplify(4^(1/2)+3);
```

$$5$$

```
> simplify(exp(a+ln(b*exp(c))));
```

$$b\,e^{(a+c)}$$

```
> simplify(sin(x)^2+cos(x)^2);
```

$$1$$

```
> simplify(sin(x)^2);
```

$$1-\cos(x)^2$$

```
> e := cos(x)^5 + sin(x)^4 + 2*cos(x)^2 - 2*sin(x)^2 - cos(2*x):  simplify(e);
```

$$\cos(x)^4\,(\cos(x)+1)$$

```
> g := sqrt(x^2);
```

$$g := \sqrt{x^2}$$

```
> simplify(g,assume=real);
```

$$|x|$$

Ακόμα μια χρήσιμη εντολή είναι η εντολή ***factor***. Με αυτή την εντολή μπορούμε να παραγοντοποίησουμε μια παράσταση.

Παραδείγματα:

```
> factor(x^2+2*x+1);
```

$$(x+1)^2$$

Στο επόμενο παράδειγμα δεδομένου του πολυωνύμου f, θέλουμε να παραγοντοποιήσουμε το πολυώνυμο $f-1$.

```
> f:=(x+y+1)^3:  factor(f-1);
```

$$(x+y)\,(x^2+2\,x\,y+3\,x+3+3\,y+y^2)$$

Στο επόμενο παράδειγμα θεωρούμε τα πολυώνυμα p_1 και p_2.

```
> p1 := -3*x+7*x^2-3*x^3+7*x^4;
```

$$p1 := -3\,x + 7\,x^2 - 3\,x^3 + 7\,x^4$$

```
> p2 := 5*x^5+3*x^3+x^2-2*x+1;
```

$$p2 := 5\,x^5 + 3\,x^3 + x^2 - 2\,x + 1$$

Θέλουμε να παραγοντοποιήσουμε το γινόμενο των πολυωνύμων p_1p_2.

```
> factor(p1*p2);
```

$$x\,(7\,x-3)\,(1+x^2)^2\,(5\,x^3-2\,x+1)$$

Εδώ θεωρούμε τα πολυώνυμα f και g.

```
> f := x^5-x^4-7*x^3+x^2+6*x:  factor(f);
```

$$x\,(x-1)\,(x-3)\,(x+2)\,(x+1)$$

```
> g := x^4+4*x^3*y-7*x^2*y^2-22*x*y^3+24*y^4:  factor(g);
```

$$(-y+x)\,(-2\,y+x)\,(3\,y+x)\,(4\,y+x)$$

```
> r := f/g:  factor(r);
```

$$\frac{x\,(x-1)\,(x-3)\,(x+2)\,(x+1)}{(-y+x)\,(-2\,y+x)\,(3\,y+x)\,(4\,y+x)}$$

Μερικές φορές θέλουμε να κάνουμε αντικατάσταση σε μια παράσταση. Σε αυτή την περίπτωση χρησιμοποιούμε την εντολή ***subs***.

Παραδείγματα:

```
> subs( x=2, x^2+x+1 );
```

$$7$$

```
> expr:= x*y+y^2:  subs(x=3,expr);
```

$$3\,y + y^2$$

```
> y := 3*x^2+4*x+5:  subs(x=1,y);
```

$$12$$

```
> subs( sin(x)=z, sin(x)/sqrt(1-sin(x)) );
```

$$\frac{z}{\sqrt{1-z}}$$

Η Maple έχει την δυνατότητα να υπολογίζει παραγώγους παραστάσεων. Η εντολή είναι ***diff***.

Παραδείγματα:

```
> diff(x^4-3*x^3+2*x^2-5*x+7,x);
```

$$4\,x^3 - 9\,x^2 + 4\,x - 5$$

```
> diff(exp(-x^2),x);
```

$$-2\,x\,e^{(-x^2)}$$

```
> diff(sqrt(x^3+1),x);
```

$$\frac{3\,x^2}{2\,\sqrt{x^3+1}}$$

Στο επόμενο παράδειγμα θέλουμε να βρούμε την δεύτερη παράγωγο ως προς x της συνάρτησης f.

```
> diff((x^2-2)*exp(-x^2/2),x,x):  simplify(%);
```

$$e^{(-\frac{x^2}{2})}\,(4 - 7\,x^2 + x^4)$$

Για να βρούμε την δεύτερη παράγωγο μιας συνάρτησης, μπορούμε να χρησιμοποιήσουμε και τον πιο κάτω τρόπο.

```
> f := sin(x)^2*cos(y):  diff(f,x$2);
```

$$2\cos(x)^2\cos(y) - 2\sin(x)^2\cos(y)$$

Τώρα θέλουμε να βρούμε την παράγωγο $\frac{d^2 f}{dxdy}$.

```
> diff(f,x,y);
```

$$-2\sin(x)\sin(y)\cos(x)$$

Πιο κάτω θα υπολογίσουμε την παράγωγο $\frac{d^3 f}{dxd^2 y}$.

```
> diff(f,x,y$2);
```

$$-2\sin(x)\cos(y)\cos(x)$$

Επίσης με την χρήση της Maple μπορούμε να ολοκληρώσουμε μια παράσταση με την εντολή ***int*** ή ***Int***.

Παραδείγματα:

```
> i1 := Int(x^2+x-3,x):  i1=value(i1);
```

$$\int x^2 + x - 3\,dx = \frac{1}{3}\,x^3 + \frac{1}{2}\,x^2 - 3\,x$$

```
> i2 := Int(sqrt(x-1)/x,x):  i2=value(i2);
```

$$\int \frac{\sqrt{x-1}}{x}\,dx = 2\,\sqrt{x-1} - 2\arctan(\sqrt{x-1})$$

```
> int((1+x)^2,x);
```

$$\frac{(1+x)^3}{3}$$

```
> int(sqrt(exp(x)-1),x);
```

$$2\sqrt{e^x-1}-2\arctan(\sqrt{e^x-1})$$

Πιο κάτω θα υπολογίσουμε ορισμένα ολοκληρώματα.

```
> i3 := Int(x*exp(b*x^2),x=0..1):  i3=value(i3);
```

$$\int_0^1 x\,e^{(b\,x^2)}\,dx=\frac{1}{2}\,\frac{e^b-1}{b}$$

```
> i4 := Int(1/(1+x^3),x=1/2..1):  i4=value(i4);
```

$$\int_{1/2}^1 \frac{1}{1+x^3}\,dx=\frac{\sqrt{3}\,\pi}{18}+\frac{1}{3}\ln(2)-\frac{1}{6}\ln(3)$$

```
> i5 := Int(1/((1+x)*(1+x^2)),x=0..infinity):  i5=value(i5);
```

$$\int_0^\infty \frac{1}{(1+x)\,(1+x^2)}\,dx=\frac{\pi}{4}$$

```
> int(x^3*exp(x),x=-2..1);
```

$$-2\,e+38\,e^{(-2)}$$

Επίσης με το πρόγραμμα Maple μπορούμε να λύσουμε διαφορικές εξισώσεις,χρησιμοποιώντας την εντολή ***dsolve***. Σ' αυτή την εντολή μπορούμε να δώσουμε και τις αρχικές συνθήκες του προβλήματος.

Παράδειγμα:

```
> restart;
> eq:=diff(x(t),t)+alpha*x(t)=0;
```

$$eq\;:=\;\frac{d}{dt}x\,(t)+\alpha\,x\,(t)=0$$

```
> dsolve(eq,x(0)=x0,x(t));
```

$$x\,(t)=x0\,e^{-\alpha\,t}$$

Το πρόγραμμα Maple μπορεί να χρησιμοποιηθεί στην ανάπτυξη σειρών. Αυτό πετυχαίνετε με την εντολη ***series***.

Παραδείγματα:

```
> series(sin(x)/x,x);
```

$$series\left(1-1/6\,x^2+\frac{1}{120}\,x^4+O\left(x^5\right),x,5\right)$$

```
> series(exp(x),x=delta,3);
```

$$series\left(e^{\delta}+e^{\delta}(x-\delta)+1/2\,e^{\delta}(x-\delta)^2+O\left(x-\delta^3\right),x-\delta,3\right)$$

Με την εντολή ***solve*** μπορούμε να λύσουμε μια εξίσωση ως προς μια μεταβλητή ή μπορούμε να λύσουμε ένα σύστημα.

Παραδείγματα:

```
> z := {a*x+b*y=A,c*x+d*y=B}:  solve(z,{x,y});
```

$$\{y=\frac{a\,B-c\,A}{a\,d-c\,b},\,x=-\frac{b\,B-A\,d}{a\,d-c\,b}\}$$

```
> solve({x^3-6*x^2+11*x-6=0},{x});
```

$$\{x=1\},\,\{x=2\},\,\{x=3\}$$

```
> solve({3*x+2*y=2,2*x-2*x*y+y=1},{x,y});
```

$$\{x=0,\,y=1\},\,\{x=\frac{1}{2},\,y=\frac{1}{4}\}$$

```
> solve(a+ln(x-3)-ln(x),x);
```

$$\frac{3\,e^a}{-1+e^a}$$

```
> eq := x^4-5*x^2+6*x=2:  solve(eq,x);
```

$$1,\,1,\,-1+\sqrt{3},\,-1-\sqrt{3}$$

Στο επόμενο παράδειγμα θεωρούμε το παρακάτω σύστημα:

```
> eqns := {u+v+w=1, 3*u+v=3, u-2*v-w=0};
```

$$eqns:=\{u+v+w=1,\,3\,u+v=3,\,u-2\,v-w=0\}$$

```
> solve(eqns);
```

$$\{u=\frac{4}{5},\,w=\frac{-2}{5},\,v=\frac{3}{5}\}$$

Όταν σε ένα πολυώνυμο δώσουμε την εντολή ***collect***, τότε το πολυώνυμο γράφεται κατά τις φθίνουσες δυνάμεις της μεταβλητής που επιλέγουμε.

Παραδείγματα:

```
> restart;
> p:=expand((1+x-a)*(1-x)^2);
```

$$p\,:=\,1-x-x^2+x^3-a+2\,ax-ax^2$$

```
> collect(p,x);
```

$$x^3+(-1-a)\,x^2+(-1+2\,a)\,x+1-a$$

```
> collect(p,a);
```

$$\left(-1+2\,x-x^2\right)a+1-x-x^2+x^3$$

Στην πρώτη περίπτωση το πολυώνυμο είναι διατεταγμένο ως προς τις φθίνουσες δυνάμεις του x ενώ στη δεύτερη ως προς τις δυνάμεις του a.

Για να κατατάξουμε το πολυώνυμο ως προς τις αύξουσες δυνάμεις μιας μεταβλητής, χρησιμοποιούμε την εντολή ***sort***.

Παράδειγμα:

```
> restart;
> p:=expand((1+2*x)*(a+x)^2);
```

$$p := a^2+2\,xa^2+2\,xa+4\,x^2a+x^2+2\,x^3$$

```
> sort(p,x);
```

$$a^2+2\,xa^2+2\,xa+4\,x^2a+x^2+2\,x^3$$

Μερικές φορές θέλουμε να αποσπάσουμε ένα μέρος από μια παράσταση. Για παράδειγμα, να βρούμε από ένα πολυώνυμο τον συντελεστή μιας μεταβλητής που εμφανίζεται σε αυτό. Αυτό το πετυχαίνουμε με την εντολή ***coeff*** .

Παραδείγματα:

```
> p1 := 5*x^6+3*x^4+x^2-2*x+3;
```

$$p1 := 5\,x^6+3\,x^4+x^2-2\,x+3$$

```
> p2 := 3*x^2-4*x^4+x^3-13*x+1;
```

$$p2 := 3\,x^2-4\,x^4+x^3-13\,x+1$$

Πιο κάτω θέλουμε απο το πολυώνυμο p_1 τους συντελεστές του x^3 και x^4, ενώ από το πολυώνυμο p_2 τους συντελεστές του x^2 και τον σταθερό όρο.

```
> coeff(p1,x^3); coeff(p1,x,4); coeff(p2,x,2); coeff(p2,x,0);
```

$$0$$

$$3$$

$$3$$

Πιο κάτω θέλουμε τους συντελεστές των διαφόρων δυνάμεων του x που εμφανίζονται στο πολυώνυμο p_1.

```
> coeffs(p1,x,'powers');powers;
```

$$1,\ -13,\ 1,\ -4,\ 3$$

$$1,\ x,\ x^3,\ x^4,\ x^2$$

Πιο κάτω θέλουμε τους συντελεστές των διαφόρων δυνάμεων του x που εμφανίζονται στο πολυώνυμο p_2.

```
> coeffs(p1,x,'powers');powers;
```

$$3,\ -2,\ 5,\ 3,\ 1$$

$$1,\ x,\ x^6,\ x^4,\ x^2$$

Για να φτιάξουμε πίνακα στο Maple χρησιμοποιούμε την εντολή ***array*** ή ***matrix***.

Παραδείγματα:
Θεωρούμε τους πίνακες m_1 και m_2. Θα υπολογίσουμε το $m_1 + m_2,\ m_1^{-1},\ 2m_2 - m_1$ και $m_1 m_2$.

```
> m1 := array([[1,0,2],[2,-1,3],[4,1,8]]);
```

$$m1 := \begin{bmatrix} 1 & 0 & 2 \\ 2 & -1 & 3 \\ 4 & 1 & 8 \end{bmatrix}$$

```
> m2 := array([[-3,2,1],[1,0,1],[2,7,4]]);
```

$$m2 := \begin{bmatrix} -3 & 2 & 1 \\ 1 & 0 & 1 \\ 2 & 7 & 4 \end{bmatrix}$$

```
> evalm(m1+m2);
```

$$\begin{bmatrix} -2 & 2 & 3 \\ 3 & -1 & 4 \\ 6 & 8 & 12 \end{bmatrix}$$

```
> evalm(m1^(-1));
```

$$\begin{bmatrix} -11 & 2 & 2 \\ -4 & 0 & 1 \\ 6 & -1 & -1 \end{bmatrix}$$

```
> evalm(2*m2-m1);
```

$$\begin{bmatrix} -7 & 4 & 0 \\ 0 & 1 & -1 \\ 0 & 13 & 0 \end{bmatrix}$$

```
> evalm(m1 &* m2);
```

$$\begin{bmatrix} 1 & 16 & 9 \\ -1 & 25 & 13 \\ 5 & 64 & 37 \end{bmatrix}$$

Στα πιο κάτω παραδείγματα θα χρησιμοποιήσουμε την εντολή matrix. Έστω οι πίνακες:

$$A := \begin{bmatrix} 1 & -1 & 2 \end{bmatrix}, \quad B := \begin{bmatrix} 3 \\ 2 \\ 1 \end{bmatrix}, \quad C := \begin{bmatrix} 1 & -2 & 4 \\ 2 & 0 & 3 \\ 1 & 2 & 1 \end{bmatrix}.$$

Θα υπολογίσουμε (όπου ορίζονται) τους πιο κάτω πίνακες: AB, CA, CA^T, B^TC, CC^T, AC^TB. Για να το πετύχουμε αυτό θα πρέπει να χρησιμοποιήσουμε το πακέτο *linalg*.

```
> with(linalg):
> A := Matrix ([1, -1, 2]);
```

$$A := \begin{bmatrix} 1 & -1 & 2 \end{bmatrix}$$

```
> B := Matrix([[3], [2], [1]]);
```

$$B := \begin{bmatrix} 3 \\ 2 \\ 1 \end{bmatrix}$$

```
> C := Matrix([[1,-2,4],[2,0,3],[1,2,1]]);
```

$$C := \begin{bmatrix} 1 & -2 & 4 \\ 2 & 0 & 3 \\ 1 & 2 & 1 \end{bmatrix}$$

```
> At := linalg[transpose](Matrix ([A]));
```

$$At := \begin{bmatrix} 1 \\ -1 \\ 2 \end{bmatrix}$$

```
> Bt := linalg[transpose](Matrix([B]));
```

$$Bt := \begin{bmatrix} 3 & 2 & 1 \end{bmatrix}$$

```
> Ct := linalg[transpose]( Matrix([C]));
```

$$Ct := \begin{bmatrix} 1 & 2 & 1 \\ -2 & 0 & 2 \\ 4 & 3 & 1 \end{bmatrix}$$

```
> evalm(A &* B);
```

$$\begin{bmatrix} 3 \end{bmatrix}$$

```
> evalm(C &* A);

Error, (in linalg[multiply]) non matching dimensions for
vector/matrix product
```

```
> evalm(C &* At);
```

$$\begin{bmatrix} 11 \\ 8 \\ 1 \end{bmatrix}$$

```
> evalm(Bt &* C);
```

$$\begin{bmatrix} 8 & -4 & 19 \end{bmatrix}$$

```
> evalm(C &* Ct);
```

$$\begin{bmatrix} 21 & 14 & 1 \\ 14 & 13 & 5 \\ 1 & 5 & 6 \end{bmatrix}$$

```
> evalm(A &* Ct &* B);
```

$$\begin{bmatrix} 50 \end{bmatrix}$$

Επίσης με την βοήθεια του Maple μπορούμε να υπολογίσουμε όρια συναρτήσεων χρησιμοποιώντας την εντολή ***limit*** ή ***Limit***.

Παραδείγματα:
Πρώτα ορίζουμε την συνάρτηση και μετά θα υπολογίσουμε τα όρια.

```
> y := (x^3+8)/(x+2);
```

$$y := \frac{x^3 + 8}{x + 2}$$

```
> limit(y, x=-2);
```

```
> y := sqrt(x^2+16);
```

$$y := \sqrt{x^2+16}$$

```
> limit(y, x=-3);
```

$$5$$

Διαφορετικά μπορούμε να βρίσκουμε τα όρια ως ακολούθως:

```
> Limit(ln(x)/x, x=infinity):  % = value(%);
```

$$\lim_{x\to\infty} \frac{\ln(x)}{x} = 0$$

```
> Limit((1-cos(x))/x, x=0):  % = value(%);
```

$$\lim_{x\to 0} \frac{1-\cos(x)}{x} = 0$$

```
> Limit(cos(x)^(1/x^3), x=0, 'left'):  % = value(%);
```

$$\lim_{x\to 0-} \cos(x)^{\left(\frac{1}{x^3}\right)} = \infty$$

```
> Limit(cos(x)^(1/x^3), x=0, 'right'):  % = value(%);
```

$$\lim_{x\to 0+} \cos(x)^{\left(\frac{1}{x^3}\right)} = 0$$

Η εντολή που χρησιμοποιούμε για να αναπαραστήσουμε γραφικά μια παράσταση στο επίπεδο είναι ***plot***. Ενώ για να παραστήσουμε μια γραφική παράσταση σε τρείς διαστάσεις χρησιμοποιούμε την εντολή ***plot3d***. Πιο κάτω θα δώσουμε μερικά παραδείγματα.

Παραδείγματα:

```
> plot(x*sin(x), x=-10..10);
```




```
> p2:=plot(x^3, x=-2..2, style=line):
> plots[display]({p1,p2});
```



> plot3d(x*sin(Pi*x*y),x=-1..1,y=0..1);



> plot3d((x^2+(y+1)^2)*(x^2+(y-1)^2),x=-2..2,y=-2..2);



```
> plot3d(sqrt(abs(x*y)),x=-2..2,y=-2..2,axes=boxed);
```



Ασκήσεις

1. Με τις κατάλληλες εντολές μετασχηματίστε την μια παράσταση στην άλλη και αντίστροφα:

$$
\begin{array}{ll}
a)\ x+y+\dfrac{1}{x+y}, & \dfrac{(x+y)^2+1}{x+y} \\
b)\ e^{x+y}, & e^x e^y \\
c)\ \ln(\dfrac{x}{y}), & \ln x - \ln y \\
d)\ x^{y+z}, & x^y x^z \\
e)\ \sqrt{x^2-1}, & \sqrt{x-1}\sqrt{x+1} \\
f)\ \dfrac{x}{x^2+x}+\dfrac{2}{x^2+x}+\dfrac{1}{(x^2+x)x}, & \dfrac{(x+1)^2}{(x^2+x)x}.
\end{array}
$$

2. Να υπολογίσετε απλοποίησετε τις παρακάτω παραστάσεις:

$$
\begin{array}{ll}
a)\ \dfrac{e^x+x}{e^{2x}+2xe^x+x^2}, & b)\ \dfrac{(x-2)^{3/2}}{(x^2-4x+4)^{1/4}} \\
c)\ \dfrac{\sqrt{x}-y}{x-y^2}, & d)\ \dfrac{1}{2+5^{1/3}} \\
e)\ \cos(x+y)+\sin x \sin y + 2^{x+y}, & f)\ 2\cos^2 x - \cos 2x \\
f)\ \dfrac{x^4+x^3-4x^2-4x}{x^4+x^3-x^2-2}. &
\end{array}
$$

3. Να παραγοντοποιήσετε τα πολυώνυμα:

$$
\begin{array}{l}
a)\ x^2-x,\ \ \text{στο χώρο } Z_2 \\
b)\ x^7+x^5+2x^3+2x^2+3x+2,\ \ \text{στο χώρο } Z_5,\ \ Z_7 \\
c)\ 2x^4-3x^2+x+4,\ \ \text{στο χώρο } Z_7 \\
d)\ 2x^5-6x^4-4x^3+x^2-3x-2,\ \ \text{στο χώρο } Z_7 \\
e)\ \ x^{23}-x,\ \ \text{στο χώρο } Z_{23}.
\end{array}
$$

4. Να βρείτε τις τιμές των παραστάσεων:

$$
\begin{array}{l}
a)\ 3x^2+4x+5,\ \ \text{στο } x=1 \\
b)\ 9x^3+5x^2+2x+10,\ \ \text{στο } x=1,\ \dfrac{4}{3},\ 2.45 \\
c)\ 3x^2y+y^2-4xy,\ \ \text{στο } x=1,\ y=-3.
\end{array}
$$

5. Να βρείτε τις παραγώγους των πιο κάτω παραστάσεων:

$$
\begin{array}{l}
a)\ f(x,y)=e^{x^2+y^2},\ \ \dfrac{d^2f}{dxdy},\ \dfrac{d^4f}{dx^2dy^2} \\
b)\ x^2+y^2=c,\ \ \dfrac{d^2y}{dx^2} \\
c)\ h(x,y,z)=\dfrac{1}{x^2+y^2+z^2},\ \ \dfrac{dh}{dz},\ \dfrac{d^2h}{dx^2},\ \dfrac{d^2h}{dxdy} \\
d)\ g(x)=x^n e^{\sin x},\ \ \dfrac{dg}{dx}\ \text{στο}\ \dfrac{\pi}{6},\ \dfrac{d^2g}{dx^2} \\
e)\ \sin\sqrt{x^2+a^2},\ \ 1^{\eta}-\text{παράγωγο} \\
g)\ \sin^3 x,\ \ \ 5^{\eta}-\text{παράγωγο}
\end{array}
$$

6. Να βρείτε τις παραγώγους των πιο κάτω συναρτήσεων μέχρι δεύτερης τάξης:

$$a)\ f(x,y) = \frac{\ln(1+x^4+y^4)}{\sqrt{x^2+y^2}}$$
$$b)\ g(x,y,z) = \frac{z}{x^2+y^2+z^2}$$
$$c)\ h(x,y,z) = \frac{1}{\sqrt{(x-a)^2+(y-b)^2+(z-c)^2}}.$$

7. Να υπολογίσετε τα πιο κάτω ολοκληρώματα:

$$a)\ \int \frac{2}{\sqrt{4+x^2}dx}, \qquad b)\ \int \sqrt{(1-cx^2)^3}dx$$
$$c)\ \int \frac{1}{x^4-1}dx, \qquad d)\ \int \frac{1}{x^4-4}dx$$
$$e)\ \int \sin 3x \cos 2x dx, \qquad f)\ \int xe^{ax^2}dx$$
$$g)\ \int x^{10}e^x dx, \qquad h)\ \int_0^\infty \frac{1}{(1+x)(1+x^2)}dx$$
$$i)\ \int_0^{1/2} \frac{1+x+x^2}{1-x+x^2}dx, \qquad j)\ \int_0^{\pi/6} \frac{1}{\cos x + \cos^3 x}dx$$
$$k)\ \int_0^1 \frac{1}{(x+1)\sqrt{1+x^2}}dx, \qquad l)\ \int_0^{\pi/2} \frac{1}{1-2a\cos x + a^2}dx, \quad a \neq 1$$

8. Να λύσετε την διαφορική εξίσωση

$$\frac{dy}{dx} = \sin(\pi y \sin x), \quad \text{με} \quad y(0) = a.$$

9. Να λύσετε την διαφορική εξίσωση

$$\frac{d^2x}{dt^2} + tx^3 = 0, \quad \text{με} \quad x(0) = 0, \quad \dot{x}(0) = v > 0.$$

10. Να αναπτυχθεί σε σειρά η συνάρτηση:

$$y = \arctan x \quad \text{γύρω από το} \quad x = 0.$$

11. Να αναπτυχθεί σε σειρά η συνάρτηση:

$$y = x^{x^x} \quad \text{γύρω από το} \quad x = 0.$$

12. Να αναπτυχθεί σε σειρά η συνάρτηση:

$$y = \frac{1}{1+\sqrt{x}} \quad \text{γύρω από το} \quad x = 0.$$

13. Να λύσετε τις πιο κάτω εξισώσεις/συστήματα:

$a)$ $(x-1)(x^2+x+1) = 0$
$b)$ $w+x+y+z = 1, \quad w+y = 0, \quad 2w+z = 2, \quad v+z = 0$
$c)$ $(x+1)^{(x+a)} = (x+1)^2$, ως προς x
$d)$ $x+y = 2, \quad -2x+3y = 5$

$e)\ 3x+y+w=1, \quad x+3y+w=8, \quad x+y+3w=1$
$f)\ x^2+y^2=25, \quad y=x^2-5$
$g)\ \arccos 3x = 2arcsinx,$ ως προς x
$h)\ |x+|x+1||=1,$ ως προς x
$i)\ (a^2-1)x^2+(2a^2+a-3)x+2a-2=0,$ ως προς x.

14. Να αναπτυχθεί το πολυώνυμο
$$(x+y+z)^2(x+y+1)$$
και να γραφτεί κατά τις φθίνουσες δυνάμεις α) του x, β) του z, γ) του y.
15. Να αναπτυχθεί το πολυώνυμο
$$(x+a\sqrt(x)+bx^{1/3})^3+(x+b\sqrt{x})^2$$
και να το διατάξετε κατά τις αύξουσες δυνάμεις x.
16. Να αναπτυχθεί το πολυώνυμο
$$(x+y+1)^3(x+3y+5)$$
και να το διατάξετε κατά τις αύξουσες δυνάμεις x.
17. Αν
$$f=(a_0+a_1\sin x+a_2\sin 2x+a_3\sin 3x)^4$$
να βρείτε τους συντελεστές του $\sin kx$ και $\cos kx$ για $k=1,2,3$.
18. Αν
$$f(x,y)=(x+y+1)^5(x^2+y+2)^2(x+y^2+3)$$
να βρείτε τους συντελεστές των x^3 και x^5.
19.Αν
$$A=\begin{bmatrix}1&2&3\\2&3&0\\4&8&5\end{bmatrix}, \quad B=\begin{bmatrix}3&8&-2\\4&7&-1\\0&3&5\end{bmatrix}$$
να υπολογίσετε τα $A-B,\ 2B^2-A,\ A^{-1},\ 3BA-4A^3,\ A^tB$.
20.Αν
$$A=\begin{bmatrix}1&0&2\\2&-1&3\\4&1&8\end{bmatrix}, \quad B=\begin{bmatrix}-3&2\\0&1\\7&4\end{bmatrix}$$
να υπολογίσετε τα $A^{-1},\ AA^t,\ B^tAB,\ (2A+BB^t)A^t$.
21. Να υπολογίσετε τα πιο κάτω όρια

$a)\ \lim_{x\to 0}\left(\frac{a\sin bx - b\sin ax}{x^3}\right)$

$b)$ για $y=\left(\frac{x}{1+x^2}\right)^{x/(1-x)}$, να βρείτε τα $\lim_{x\to 0} y,\ \lim_{x\to 0} y'$

$c)\ \lim_{x\to\infty}(1+\frac{\pi}{x})^x$

$d)\ \lim_{x\to 0} x^{\sin x}$

$e)\ \lim_{x\to\infty}(2^x+3^x)^{1/x}$

$f)\ \lim_{x\to\infty}\frac{\ln x}{x}$

$g)\ \lim_{x\to\infty}\frac{\ln x}{e^x}$

$h)\ \lim_{x\to\infty} \dfrac{x^2+\sin x}{2x^2+\cos 4x}$

$i)\ \lim_{x\to 0} \dfrac{2}{1+e^{-1/x}}.$

22. Να παραστήσετε γραφικά τις πιο κάτω συναρτήσεις:

$a)\ y = 15x^5 - 15x^4 + 20x^3 - 330x^2 + 600x + 2, \quad 0 < x < 3$

$b)\ y = e^x + ln|4-x|, \quad 0 \le x \le 5$

$c)\ z(x,y) = x(x^2-3y^2), \quad -1 \le x, \quad y \le 1$

$d)\ y^3 + x^3 - 9xy + 1 = \pm 1, \quad -2 < x < 5, \quad -2 < y < 5.$